

Modeling a Simple Verilog Cell Library

Christian Krieg

October 29, 2024

Welcome to this year's *Digital Integrated Circuits* course! Throughout this course, we will have to solve some assignments that will teach us fundamental concepts of integrated circuit design.

We will demonstrate digital integrated circuit design on a step-by-step simple example of a digital counter. As the name of this course suggests, our ultimate goal is to implement this counter as a digital integrated circuit. In the lab, we map this counter to a [field-programmable gate array \(FPGA\)](#) architecture, and run it on the board shown in Figure 1. The counter's functionality is executed on the [FPGA](#). We will be able to control execution by pushing the buttons, and to visualize the counter value using the [light-emitting diodes \(LEDs\)](#). Because operating a hardware counter with buttons and [LEDs](#) may be inconvenient, we also will interface the hardware counter over a [universal asynchronous receiver/transmitter \(UART\)](#), so that we may be able to further process the counter value on a host computer. But this will all happen during the lab week at the end of the exercises.




Figure 1: The *icebreaker* board, which will run our counter

Before we can bring our counter on hardware, it is important that we learn how to design digital circuits using state-of-the-art (and beyond) methods and tools, and to get a feeling for the results of our design decisions. We will draw attention on the results of the synthesis steps, which we will verify with several tools along the digital design flow.

Generally speaking, we perform the following steps:

1. Specify the counter's behavior
2. **Model (implement) the counter in a [hardware description language \(HDL\)](#)**
3. **Verify the counter's implementation**
4. Synthesize and optimize the counter's [HDL](#) model + verify
5. Map the counter to the target ([FPGA](#)) technology + verify
6. Generate an [FPGA](#) bitstream + verify
7. Configure the [FPGA](#) with the bitstream + verify
8. Run the counter on hardware

Our overall task is to build a synchronous counter that runs in two modes, based on a *mode* switch. The two modes are called *up* and *down*. In *up* mode, the counter increments the counter value by **5** at the rising edge of the clock signal, and in *down* mode, it decrements the counter value by **9** at the rising edge of the clock signal. The counter should implement synchronous reset behavior, such that it is set to an initial value of **-50** upon reset. The counter should never exceed a value of **235**, and should never go below a value of **-230** (the counter value sticks near the minimum/maximum value, until the counting direction changes). The counter value should never be **-11** (the counter should jump over this value by one increment/decrement).

Wrapping up Task 3

We hope that you enjoyed to model your counter in behavioral Verilog, and that your assertions helped you to debug and verify your counter's implementation. In case you did not come up with a working solution, we provide a working solution in Listing 1 which may be used for future tasks.

Listing 1: Sample solution for counter in Verilog (*counter.v*)

```
1 module counter (clk, rst, mode, cnt);
2
3     input wire clk;
4     input wire rst;
5     input wire mode;
6     output reg signed [9:0] cnt = -10'sd50;
7
8     reg signed [9:0] summand;
9
10    always @* begin
11        summand = 10'sd0;
12        if (mode) begin
13            if (cnt != -10'sd16) begin
14                if (cnt <= 10'sd230) begin
15                    summand = 10'sd5;
16                end
17            end else if (cnt <= 10'sd225) begin
18                summand = 10'sd10;
19            end
20            // mode = 0
21        end else begin
22            if (cnt != -10'sd2) begin
23                if (cnt >= -10'sd221) begin
24                    summand = -10'sd9;
25                end
26            end else if (cnt >= -10'sd212) begin
27                summand = -10'sd18;
28            end
29        end
30    end
31
32    always @(posedge clk) begin
33        if (rst)
34            cnt <= -10'sd50;
35        else begin
36            cnt <= cnt + summand;
37        end
38    end
39
40    `ifdef FORMAL
41        reg init = 1;
42        always @(posedge clk) begin
43            if (init) assume(rst);
44            else assume(!rst);
45            init <= 0;
46        end
47
48        always @(posedge clk) begin
49
50            if (rst) begin
51                assert (cnt == -10'sd50);
52            end
53
54            if (!rst) begin
55
56                //
57                // Check if counter value is never lower than MIN, larger than MAX, or
58                // equal to INV
59                //
60                assert (cnt <= 10'sd235);
61                assert (cnt >= -10'sd230);
62                assert (cnt != -10'sd11);
63
64                //
65                // Check if the counter value is correctly incremented and decremented
66                //
67
68                // Counting up
69                if ($past(mode)) begin
70                    if (!$past(rst)) begin
71                        if ($past(cnt) == -10'sd16) // INV - INC, counter jumped over INV
72                            assert ((cnt - $past(cnt)) == 10'sd10);
73                        else if ($past(cnt) <= 10'sd235 && $past(cnt) > 10'sd230) // cnt was near MAX
74                            assert ((cnt - $past(cnt)) == 0);
75                        else
76                            assert ((cnt - $past(cnt)) == 10'sd5);
77                    end
78                end
79
80                // Counting down
81                if (!$past(mode)) begin
```

```

82         if (!$past(rst)) begin
83             if ($past(cnt) == -10'sd2) // INV + DEC, counter jumped over INV
84                 assert (($past(cnt) - cnt) == 10'sd18);
85             else if ($past(cnt) >= -10'sd230 && $past(cnt) < -10'sd221) // cnt was near MIN
86                 assert (($past(cnt) - cnt) == 0);
87             else
88                 assert (($past(cnt) - cnt) == 10'sd9);
89         end
90     end
91 end
92 end
93 end
94 'endif
95 endmodule

```

As we can see, we use one addition instead of multiple adders and/or subtractors. Also, we do not host multipliers in our design. Multipliers are expensive in terms of area, and the data range is different which results in a different bit width for the output signals.

Note that we put all combinational logic into an `always @*`-block, and use *blocking* assignments (`... = ...`). Likewise, we put all sequential logic into an `always @(posedge clk)`-block, and use *non-blocking* assignments (`... <= ...`). Instead of evaluating the counter value and setting its new value in one sequential block (which may create multiple adders/subtractors), we define an intermediate signal `summand` to hold the difference to the counter's next value, which is calculated according to the counter's current value and the status of `mode`. The value of `summand` is added to the counter value at each positive edge of the `clk` signal. Note that the counter and the `summand` signal is *signed*, and that we explicitly declare the constants' representations to be 10-bit signed decimals. On top, we carry the counter's formal specification along with its implementation, which simplifies the verification process (less files needed).

The take-away message here is that by choosing a suitable implementation style, we potentially save significant amounts of resources (adders and subtractors). In addition, the design is potentially clearer, and easier to maintain. Coding style matters!

Low-level modeling

For now, the (arithmetic) addition operation (“+”) in our counter is modeled by generic `$add` cells. We provide the inputs and obtain their sum at the outputs. We do not specify yet in detail how the sum is calculated. It is now time to define an underlying Boolean implementation of the adder in our design.

In this task, we learn to implement a simple N -bit [carry-lookahead adder \(CLA\)](#). Make yourself familiar with the concept of [CLA](#) in order to understand what it is used for. We will model our adder implementation in Verilog, and we will use technology mapping (*techmap*), a very practicable feature of Yosys, to substitute the generic `$add` cell by its low-level implementation.

In Verilog, functional blocks are modeled in *modules*, which may be parameterized with the construct of *parameters* (like *generics* in VHDL). A module's interface is defined using *input* and *output* directives. Like the VHDL *downto* directive, in Verilog ranges of bits are declared using brackets, with the upper bound left of a colon, and the lower-bound right of the colon (e.g., `[12:0]` means from bit 12 down to bit 0). Like VHDL, Verilog supports *generate* statements in order to instantiate multiple hardware blocks based on parameters and conditions. There are *blocking* statements (*assign*) generally used for combinational circuits, and *non-blocking* statements (`<=`), generally used for modeling sequential circuits. The equivalent of *signals* in VHDL are *wires* in Verilog.

In Listing 2, we can see a module definition of an `$add` cell type. The module is parameterized with the width of input and output signals, and the signedness of input signals. The parameters get default values, which are replaced by actual values when the modules are instantiated in an actual design. The module definitions also define input signals (*A*, *B*) and output (*Y*) signals. Note that while the widths of inputs *A* and *B* may be defined as independent values, we only consider `A_WIDTH = B_WIDTH`. Also note that the adder's output signal is one bit larger than its input signals. This is because the output carry bit is assigned to the output signal's most-significant bit.

Listing 2: Skeleton Verilog module to implement `$add` cells (*map.v*)

```

1 //-----
2 //
3 // Maps $add cell to a carry-lookahead adder (CLA) implementation
4 //
5 module \ $add (A, B, Y);
6
7     parameter A_WIDTH = 32;
8     parameter B_WIDTH = 32;
9     parameter Y_WIDTH = 32;
10    parameter A_SIGNED = 0;
11    parameter B_SIGNED = 0;
12

```

```

13|     input  [A_WIDTH-1:0] A;
14|     input  [B_WIDTH-1:0] B;
15|     output [Y_WIDTH:0] Y;
16|
17|     // Implement CLA functionality here
18|
19| endmodule
20| //
21| //-----

```

Technology mapping

Yosys provides a very practicable feature that allows us to replace cells by their low-level implementation. It is called *technology mapping*, and implemented in the *techmap* command. *techmap* accepts a Verilog file as input which includes the mappings between cells and their implementation. When we provide *map.v* to the *techmap* command, *techmap* will look for *\$add* cells, and will replace them with their low-level implementation as specified by you. The command to invoke *techmap* is given in Listing 3.

Listing 3: Yosys command to replace adders by their low-level implementation

```

1| techmap -map map.v

```

We synthesize our counter as we did in Tasks 3, then we perform technology mapping as shown in Listing 3, and see what happens to our counter using the *show* and *stat* commands. We may write our counter to a Verilog file using the *write_verilog* command (as shown in Listing 4). We verify the functional correctness of our counter (given in Verilog) with *SymbiYosys* using our formal testbench. Listing 5 shows the *SymbiYosys* configuration file.

Listing 4: Yosys command to write our counter to a Verilog file

```

1| write_verilog counter-cla.v

```

Listing 5: *SymbiYosys* configuration to verify our counter's Verilog model (counter.sby)

```

1| [options]
2| mode prove
3| expect pass
4|
5| [engines]
6| aiger suprove
7|
8| [script]
9| read -formal tb_counter.sv counter-cla.v
10| verific -import tb_counter
11| prep -top tb_counter
12|
13| [files]
14| counter-cla.v
15| tb_counter.sv

```

Your task

Your task is to extend the skeleton module given in Listing 2 with *CLA* functionality for any adder bit width. Use Verilog *generate* statements. Think about how you should initialize the carry input for the adder's first bits. Use *wires* if you need to define intermediate signals. If you are unsure about the boundaries of the *generate* variable, take into consideration the output signal. If you want to access one particular bit of a signal, enclose it with brackets. For instance, if you would like to access the second bit of signal *A*, you can access it with *A[1]*.

Submission details

Attach your extended *map.v*, as well as your generated *counter-cla.v* to your submission e-mail.